Densor Package

Joshua Maglione

Universität Bielefeld

jmaglione@math.uni-bielefeld.de

James B. Wilson

Colorado State University

james.wilson@colostate.edu

Version 1.0

January 24, 2019

Contents

CHAPTER 1

Introduction

CHAPTER 2

Subspaces as closures

The subspaces in this section are given by Galois connections in [**FMW**, Theorem A]. Currently, these are the only closures that are constructible in this package, and they are implemented for 3-tensors.

```
DerivationClosure(T, Delta) : TenSpc, AlgMat -> TenSpc
DerivationClosure(T, Delta) : TenSpc, ModMatFld -> TenSpc
DerivationClosure(T, Delta) : TenSpc, AlgMatLie -> TenSpc
DerivationClosure(T, Delta) : TenSpc, [Mtrx] -> TenSpc
DerivationClosure(T, Delta) : TenSpc, [AlgMatLie] -> TenSpc
```

Returns the derivation closure of the given tensor space T, with frame $U_2 \times U_1 \rightarrowtail U_0$, with operators $\Delta \subseteq \mathrm{End}(U_2)\times\mathrm{End}(U_1)\times\mathrm{End}(U_0)$. If a nontrivial subset of coordinates are fused, then `DerivationClosure` accepts Δ input from the smaller ambient space. For example, if coordinates 2 and 1 are fused, then the input for Δ can be in either $\mathrm{End}(U_2) \times \mathrm{End}(U_1) \times \mathrm{End}(U_0)$ or $\mathrm{End}(U_2) \times \mathrm{End}(U_0)$. Currently, this only works for tensor spaces of valence 3. This is the subspace whose tensors' derivation algebra contains Δ.

```
DerivationClosure(T, t) : TenSpc, TenSpcElt -> TenSpc
```

Returns the derivation closure of the given tensor space T, with frame $U_2 \times U_1 \rightarrowtail U_0$, whose operators are the derivation algebra of t. Currently, this only works for tensor spaces of valence 3. This is the subspace whose tensors' derivation algebra contains the derivation algebra of t.

```
UniversalDensorSubspace(t) : TenSpcElt -> TenSpc
```

Returns the universal densor subspace of the given tensor $t : U_2 \times U_1 \rightarrowtail U_0$ as a subspace of the universal tensor space of t, whose operators are the derivation algebra of t. Currently, this only works for tensor spaces of valence 3. This is equivalent to `DerivationClosure(Parent(t), t)`. This tensor subspace is stored with t once computed.

Example 2.1. `Rank1Densor`

We illustrate the fact that the tensor for (hyper-)matrix multiplication spans its densor, see [**FMW**, Theorem G]. We will construct the derivation closure of $\langle t| : \mathbb{M}_{2\times 3}(\mathbb{F}_3) \times \mathbb{M}_{3\times 2}(\mathbb{F}_3) \rightarrowtail \mathbb{M}_2(\mathbb{F}_3)$, given by matrix multiplcation. The derivation closure of t in the universal tensor space T is 1-dimensional.

```
> Fr := [ KMatrixSpace(GF(3),2,3), KMatrixSpace(GF(3),3,2),
>     KMatrixSpace(GF(3),2,2) ];
> F := func< x | x[1]*x[2] >;
> t := Tensor(Fr, F);
> t;
Tensor of valence 3, U2 x U1 >-> U0
U2 : Full Vector space of degree 6 over GF(3)
U1 : Full Vector space of degree 6 over GF(3)
U0 : Full Vector space of degree 4 over GF(3)
```

The derivation algebra of t is isomorphic to $(\mathfrak{gl}_2(\mathbb{F}_3) \oplus \mathfrak{gl}_3(\mathbb{F}_3) \oplus \mathfrak{gl}_2(\mathbb{F}_3))/\mathbb{F}_3$. We do not show this, but we verify that the dimensions match.

```
> D := DerivationAlgebra(t);
> Dimension(D) eq 4+9+4-1;
true
```

Now we verify that t spans its own densor.

```
> T := Parent(t);
> T;
Tensor space of dimension 144 over GF(3) with valence 3
U2 : Full Vector space of degree 6 over GF(3)
U1 : Full Vector space of degree 6 over GF(3)
U0 : Full Vector space of degree 4 over GF(3)
> densor := DerivationClosure(T, D);
> densor eq sub< T | t >;
true
```

NucleusClosure(T, Delta, a, b) : TenSpc, AlgMat, RngIntElt, RngIntElt -> TenSpc
NucleusClosure(T, Delta, a, b) : TenSpc, ModMatFld, RngIntElt, RngIntElt -> TenSpc
NucleusClosure(T, Delta, a, b) : TenSpc, [Mtrx], RngIntElt, RngIntElt -> TenSpc

Returns the nucleus closure of the tensor space T, with frame $U_2 \times U_1 \rightarrowtail U_0$, with operators $\Delta \subseteq \mathrm{End}(U_a) \times \mathrm{End}(U_b)$. Currently, this only works for tensor spaces of valence 3. This returns the subspace whose tensors' $\{a, b\}$-nuclues contains Δ.

NucleusClosure(T, t, a, b) : TenSpc, TenSpcElt, RngIntElt, RngIntElt -> TenSpc

Returns the nucleus closure of the tensor space T, with frame $U_2 \times U_1 \rightarrowtail U_0$, whose operators are the $\{a, b\}$-nucleus of t. Currently, this only works for tensor spaces of valence 3. This returns the subspace whose tensors' $\{a, b\}$-nuclues contains the $\{a, b\}$-nucleus of t.

Example 2.2. NucClosure

We illustrate that if t is the commutator tensor from the Heisenberg group, then the densor of t is $\mathrm{Cen}(t) \cdot t$, 1-dimensional over the centroid. First, we construct t with frame $\mathbb{F}_5^6 \times \mathbb{F}_5^6 \rightarrowtail \mathbb{F}_5^3$, so that t is $\mathbb{F}_5$-bilinear.

```
> H := ClassicalSylow(GL(3,125), 5);
> t := pCentralTensor(H);
> t;
Tensor of valence 3, U2 x U1 >-> U0
U2 : Full Vector space of degree 6 over GF(5)
U1 : Full Vector space of degree 6 over GF(5)
U0 : Full Vector space of degree 3 over GF(5)
```

Because t came from the Heisenberg group over $\mathbb{F}_{5^3}$, the centroid of t will be 3-dimensional (over $\mathbb{F}_5$) and isomorphic to $\mathbb{F}_{5^3}$. After computing the centroid of t, we will rewrite t over the centroid $\mathrm{Cen}(t)$, so that t is $\mathbb{F}_{5^3}$-bilinear.

```
> C := Centroid(t);
> C;
Matrix Algebra of degree 15 and dimension 3 with 3 generators over GF(5)
> s := TensorOverCentroid(t);
> s;
Tensor of valence 3, U2 x U1 >-> U0
U2 : Full Vector space of degree 2 over GF(5^3)
U1 : Full Vector space of degree 2 over GF(5^3)
U0 : Full Vector space of degree 1 over GF(5^3)
```

Now we will compute the nucleus closure of both s and t at $\{2, 1\}$. The dimension of the closure of s will be 1-dimensional, while the dimension of the closure for t will not.

```
> NucleusClosure(Parent(s), s, 2, 1);
Tensor space of dimension 1 over GF(5^3) with valence 3
U2 : Full Vector space of degree 2 over GF(5^3)
U1 : Full Vector space of degree 2 over GF(5^3)
```

```
U0 : Full Vector space of degree 1 over GF(5^3)
>
> NucleusClosure(Parent(t), t, 2, 1);
Tensor space of dimension 9 over GF(5) with valence 3
U2 : Full Vector space of degree 6 over GF(5)
U1 : Full Vector space of degree 6 over GF(5)
U0 : Full Vector space of degree 3 over GF(5)
```

Bibliography

[FMW] Uriya First, Joshua Maglione, and James B. Wilson, *Polynomial identity tensors and their invariants*. in preparation.

Intrinsics